Linguagem Portugol

Uma linguagem de programação é um método pa
programador especifique precisamente sobre qua

O Portugol é uma representação que se assemelh

Sintaxe e semântica do Portugol

O compilador auxilia a verificar se a sintaxe e a s

Durante os tópicos da ajuda, serão apresentadas

Exemplo da estrutura básica

Exemplo

```
01. //O comando programa é obrigatório
02. programa
03. {
04.     //Inclusões de bibliotecas
05.     // - Quando houver a necessidade de
06.     //   uma ou mais bibliotecas, as in
07.     //   devem aparecer antes de qualqu
08.
09.     /*
10.      *  Dentro do programa é permitido
11.      *  variáveis globais, constantes g
12.      *  funções em qualquer ordem.
13.      */
14.
15.     //Declarações de funções somente
```

Linguagem Portugol

Uma linguagem de programação é um método padronizado para comunicar instruções para um computador. É um conjunto de regras sintáticas e semânticas usadas para definir um programa de computador. Permite que um programador especifique precisamente sobre quais dados um computador vai atuar, como estes dados serão armazenados ou transmitidos e quais ações devem ser tomadas sob várias circunstâncias.

O Portugol é uma representação que se assemelha bastante com a linguagem C, porém é escrito em português. A ideia é facilitar a construção e a leitura dos algoritmos usando uma linguagem mais fácil aos alunos.

Sintaxe e semântica do Portugol

O compilador auxilia a verificar se a sintaxe e a semântica de um programa está correta.

Durante os tópicos da ajuda, serão apresentadas as estruturas básicas da linguagem.

Exemplo da estrutura básica

Exemplo

```
01. //O comando programa é obrigatório
02. programa
03. {
04.     //Inclusões de bibliotecas
05.     // - Quando houver a necessidade de utilizar
06.     //   uma ou mais bibliotecas, as inclusões
07.     //   devem aparecer antes de qualquer declaração
08.
09.     /*
10.      *  Dentro do programa é permitido declarar
11.      *  variáveis globais, constantes globais e
12.      *  funções em qualquer ordem.
13.      */
14.
15.     //Declarações de funções somente
```

Tipos

Quais são os tipos de dados que o computador pode armazenar?

Se pararmos para pensar que tudo que o computador compreende é representado através de Zeros e Uns. Então a resposta é Zero e Um. Certo? Certo! Mas como então o computador pode exibir mensagens na tela, apresentar ambientes gráficos cheios de janelas, compreender o significado da teclas do teclado ou dos cliques do mouse.

Bom tudo começa com a definição de uma série de códigos. Por exemplo. A letra "a" do teclado é representada pela seguinte sequência de zeros e uns "01000001". O número 22 é representado por "00010110". E assim todos os dados que são armazenados pelo computador podem ser representados em zeros e ums.

Sendo assim, existem alguns tipos básicos de dados nos quais valores podem ser armazenados no computador. O Portugol exige que o tipo de dado de um valor seja do mesmo tipo da variável ao qual este valor será atribuído.

Nesta seção, serão abordados os seguintes tópicos:

- Tipo Cadeia
- Tipo Caracter
- Tipo Inteiro
- Tipo Lógico
- Tipo Real
- Tipo Vazio

060424

Tipo Inteiro

Em determinadas situações faz-se necessário a utilização de valores inteiros em um algoritmo. Como faríamos, por exemplo, uma simples soma entre dois números pertencentes ao conjunto dos números inteiros? Simples. Utilizando variáveis do tipo **inteiro**. Uma variável do tipo inteiro pode ser entendida como uma variável que contém qualquer número que pertença ao conjunto dos números inteiros. Podem ser positivos, negativos ou nulos.

A declaração de uma variável do tipo **inteiro** é simples.

A sintaxe é a palavra reservada **inteiro** e em seguida um nome para variavel

Exemplo de Sintaxe

```
01. inteiro nome_da_variavel
```

O valor que essa variável assumirá poderá ser especificado pelo programador ou solicitado ao usuário (ver Operação de Atribuição).

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         inteiro num1, num2
06.         num1 = 5
07.         num2 = 3
08.         escreva (num1 + num2)
09.     }
10. }
```

Tipo Real

Em algumas situações é necessário armazenar valores que não pertencem aos números inteiros. Por exemplo, se quiséssemos armazenar o valor da divisão de 8 por 3, como faríamos? Este problema pode ser solucionado com uma variável do tipo **real**. Uma variável do tipo **real** armazena um número real como uma fração decimal possivelmente infinita, como o número PI 3.1415926535. Os valores do tipo de dado **real** são números separados por pontos e não por virgulas.

A sintaxe para a declaração é a palavra reservada **real** junto com o nome da variável.

Exemplo de Sintaxe

```
01. real nome_da_variavel
```

O valor que essa variável assumirá poderá ser especificado pelo programador ou solicitado ao usuário (ver Operação de Atribuição).

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         real div
06.
07.         div = 8.0/3.0
08.
09.         escreva (div)
10.     }
11. }
```

Tipo Caracter

Em determinadas situações faz-se necessário o uso de símbolos, letras ou outro tipo de conteúdo. Por exemplo, em um jogo da velha, seriam necessárias variáveis que tivessem conteúdos de 'X' e '0'. Para este tipo de situação, existe a variável do tipo **caracter**. A variável do tipo caracter é aquela que contém uma informação composta de apenas UM carácter alfanumérico ou especial. Exemplos de caracteres são letras, números, pontuações e etc.

A sintaxe é a palavra reservada **caracter** e em seguida um nome para variavel

Exemplo de Sintaxe

```
01. caracter nome_da_variavel
```

O valor que essa variável assumirá poderá ser especificado pelo programador ou solicitado ao usuário (ver Operação de Atribuição). Caso seja especificado pelo programador, o conteúdo deve estar acompanhado de aspas simples.

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo de Sintaxe

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         caracter vogal, consoante
06.         vogal = 'a'                             //variável declarada através de atribuição do programa
07.
08.         escreva ("Digite uma consoante: ")
09.         leia (consoante)                        //variável declarada através de entrada do usuário
10.
11.         escreva ("Vogal: ", vogal, "\n", "Consoante: ", consoante)
12.     }
13. }
```

Tipo Cadeia

Em algumas situações precisa-se armazenar em uma variável, um texto ou uma quantidade grande de caracteres. Para armazenar este tipo de conteúdo, utiliza-se uma variável do tipo **cadeia**. Cadeia é uma sequência ordenada de caracteres (símbolos) escolhidos a partir de um conjunto pré-determinado.

A sintaxe é a palavra reservada **cadeia** seguida do nome da variavel

Exemplo de Sintaxe

```
01. cadeia nome_da_variavel
```

O valor que essa variável assumirá poderá ser especificado pelo programador, ou solicitado ao usuário (ver Operação de Atribuição). Caso seja especificado pelo programador, o conteúdo deve estar acompanhado de aspas duplas.

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         cadeia nome1, nome2
06.
07.         nome1 = "Variável declarada através de atribuição"      //variável declarada através de atribu
08.
09.         escreva ("Digite seu nome: ")
10.         leia (nome2)                                          //variável declarada através de entrada do
11.         escreva ("\nOlá ", nome2)
12.     }
13. }
```

Tipo Lógico

Em determinadas situações faz-se necessário trabalhar com informações do tipo verdadeiro e falso. Este tipo de necessidade aparece muito em operações relacionais para exibir se determinada condição é verdadeira ou falsa. Por exemplo: como poderíamos verificar se um número digitado pelo usuário é maior que zero? Através de uma variável do tipo **logico**. Uma variável do tipo **logico** é aquela que contém um tipo de dado, usado em operações lógicas, que possui somente dois valores, que são consideradas pelo Portugol como verdadeiro e falso.

A declaração de uma variável do tipo logico é simples. A sintaxe é a palavra reservada **logico** seguida do nome da variável.

Exemplo de Sintaxe

```
01. logico nome_da_variavel
```

O valor que essa variável assumirá poderá ser especificado pelo programador ou solicitado ao usuário (ver Operação de Atribuição). Lembrando que em ambos os casos a variável só assume valores verdadeiro ou falso.

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         logico teste
06.         inteiro num
07.
08.         escreva ("Digite um valor para ser comparado :")
09.         leia (num)
10.
11.         teste = (num>0)
12.
13.         escreva ("O número digitado é maior que zero? ", teste)
14.     }
15. }
```

Tipo Vazio

Vazio é usado para o resultado de uma função que retorna normalmente, mas não fornece um valor de resultado ao seu chamada.

Normalmente, essas funções de tipo **vazio** são chamados por seus efeitos colaterais, como a realização de alguma tarefa ou escrevendo os seus parâmetros na saída de dados.

A função com o tipo **vazio** termina ou por atingir o final da função ou executando um comando retorne sem valor retornado.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         imprime_linha()
06.         informacoes("Portugol",2.0,"UNIVALI")
07.         imprime_linha()
08.         informacoes("Java",1.7,"Oracle")
09.         imprime_linha()
10.         informacoes("Ruby",2.0,"ruby-lang.org")
11.         imprime_linha()
12.         informacoes("Visual Basic",6.0,"Microsoft")
13.         imprime_linha()
14.
15.     }
```

Declarações

Sejam números ou letras, nossos programas tem que conseguir armazenar dados temporariamente para poderem fazer cálculos, exibir mensagens na tela, solicitar a digitação de valores e assim por diante.

Uma declaração especifica o identificador, tipo, e outros aspectos de elementos da linguagem, tais como variáveis, constantes e funções. Declarações são feitas para anunciar a existência do elemento para o compilador.

Para as variáveis, a declaração reserva uma área de memória para armazenar valores e ainda dependendo onde ela foi declarada pode ser considerada local (vai existir somente dentro de seu escopo) ou global (vai existir enquanto o programa estiver em execução).

Para as constantes a declaração funciona de forma parecida a de uma variável, porem sem a possibilidade de alterar seu valor no decorrer do programa.

Para as funções, declarações fornecem o corpo e assinatura da função.

Nesta seção, abordaremos os seguintes tipos de declarações:

- Declaração de Constante
- Declaração de Função
- Declaração de Matriz
- Declaração de Variáveis
- Declaração de Vetor

Declaração de Variáveis

O computador armazena os dados que são utilizados nos programas e algoritmos na memória de trabalho ou memória RAM (Random Access Memory). A memória do computador é sequencial e dividida em posições. Cada posição de memória permite armazenar uma palavra (conjunto de bytes) de informação e possui um número que indica o seu endereço.

Vamos supor que queremos fazer um programa que solicita para um usuário digitar a sua idade e exibe a ele quantos anos faltam para ele atingir 100 anos de idade. Precisaremos armazenar a idade do usuário para depois realizar o cálculo 100 - idade_usuario e depois armazenar também o resultado.

Para facilitar a nossa vida de programadores, foram criadas as variáveis. As variáveis podem ser entendidas como sendo apelidos para as posições de memória. É através das variáveis que os dados dos nossos programas serão armazenados. A sintaxe para se declarar uma variável é o tipo da variável, o nome da variável ou das variáveis (separadas por virgula cada uma) e opcionalmente pode ser atribuído a ela um valor de inicialização (exceto se for declarado mais de uma na mesma linha)

Exemplo de Sintaxe

```
01. caracter nome_variavel
02. inteiro variavel_inicializada = 42
03. real nome_variavel2
04. logico nome_variavel3
05. // ou para declarar varias variáveis de um mesmo tipo:
06. cadeia var1,var2,var3,var4
07. logico var4,var5,var6
```

É importante ressaltar que o nome de cada variável deve ser explicativo, facilitando assim a compreensão do conteúdo que está armazenado nela.

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     //variável global do tipo inteiro
04.     inteiro variavel
05. 
06.     funcao inicio()
07.     {
08.         //variável local do tipo inteiro
09.         inteiro outra_variavel
10. 
11.         //variável local do tipo real já inicializada
12.         real altura = 1.79
13. 
14.         cadeia frase = "Isso é uma variável do tipo cadeia"
15. 
```

Declaração de Constante

Existem algumas situações em que precisamos que um determinado parâmetro não tenha seu valor alterado durante a execução do programa. Para isso, existem as constantes. Constante é um identificador cujo valor associado não pode ser alterado pelo programa durante a sua execução.

Para declarar uma constante basta adicionar a palavra reservada **const** seguida do tipo de dado, pelo nome da constante e atribuir um valor a ela.

Exemplo de Sintaxe

```
01. const inteiro NOME_DA_CONSTANTE = 3
02. const real NOME_DA_CONSTANTE2 = 45
```

Por uma questão de convenção, é aconselhável deixar o nome da sua constante em caixa alta (todas as letras em maiúsculo)

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     //Constante global do tipo de dado real
04.     const real aceleracao_gravidade = 9.78
05. 
06.     funcao inicio()
07.     {
08.         //Vetor constante local do tipo de dado caracter
09.         const caracter vogais[5] = {'a','e','i','o','u'}
10. 
11.         //Matriz constante local do tipo de dado inteiro
12.         const inteiro teclado_numerico[][] = {{1,2,3},{4,5,6},{7,8,9}}
13.     }
14. }
```

Declaração de Função

Se lhe fosse solicitado um algoritmo para preencher uma matriz, você o resolveria correto? Porém, se ao invés de uma matriz fossem solicitadas dez matrizes? Concordamos que o algoritmo ficaria muito cansativo e repetitivo. Mas, e se pudéssemos repetir o mesmo procedimento, quantas vezes necessário, o escrevendo apenas uma vez? Nós podemos. Para isso, usamos uma função. Função consiste em uma porção de código que resolve um problema muito específico, parte de um problema maior.

Algumas das vantagens na utilização de funções durante a programação são:

- A redução de código duplicado num programa;
- A possibilidade de reutilizar o mesmo código sem grandes alterações em outros programas;
- A decomposição de problemas grandes em pequenas partes;
- Melhorar a interpretação visual de um programa;
- Esconder ou regular uma parte de um programa, mantendo o restante código alheio às questões internas resolvidas dentro dessa função;

As componentes de uma funções são:

- O seu protótipo, que inclui os parâmetros que são passados à função na altura da invocação;
- O corpo, que contém o bloco de código que resolve o problema proposto;
- Um possível valor de retorno, que poderá ser utilizado imediatamente a seguir à invocação da função.

A declaração de função no Portugol é realizada da seguinte forma: Deve-se utilizar a palavra reservada **funcao**, seguido do tipo de retorno. Quando o tipo de retorno é ocultado, o Portugol assume que o retorno é do tipo vazio. Então, deve-se definir o nome da função seguido de abre parênteses, uma lista de parâmetros pode ser incluída antes do fecha parênteses. Para concluir a declaração deve-se criar o corpo da função. O corpo da função consiste em estruturas dentro do abre e fecha chaves. Quando uma função possui um tipo de retorno diferente de vazio, é obrigatória a presença do comando retorne no corpo da função.

Exemplo de Sintaxe

```
01. funcao real nome_da_funcao (inteiro parametro1,real parametro2)
02. {
03.     retorne parametro1 * parametro2
04. }
05. funcao inteiro nome_da_funcao2 ()
06. {
07.     retorne 1
08. }
09. funcao nome_da_funcao3 (cadeia &parametro)
10. {
11.     parametro = "Novo Valor"
12. }
```

A declaração dos parâmetros é similar a declaração de variável, vetor e matriz sem inicialização e devem ser separados por vírgula. Note que uma função do tipo vazio não tem retorno.

Para funções existem dois tipos de passagens de valores possíveis. São eles: por valor e por referência. A passagem de parâmetros por valor transfere para a função apenas o valor contido na variável, ou seja, a variável em si não terá seu conteúdo alterado. Já a passagem de parâmetro por referência transfere a variável como um todo, modificando a mesma de acordo com os comandos presentes no corpo da função.

Por padrão os parâmetros se comportam como passagem por valor, para o parâmetro se comportar como referência deve-se adicionar o simbolo & antes do nome do parâmetro.

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     //Função com retorno do tipo vazio sem parâmetro
04.     funcao vazio imprime_linha()
05.     {
06.         escreva("\n------------------------------\n")
07.     }
08. 
09.     //Função com retorno do tipo vazio e com um vetor como parâmetro
10.     funcao inicio(cadeia argumentos[])
11.     {
12.         //Imprime o retorno da função media
13.         escreva(media(4,9,8))
14. 
15.         imprime_linha()
```

Declaração de Matriz

Para a melhor compreensão do conceito de matriz, é interessante o entendimento de Vetores. Os vetores permitem solucionar uma série de problemas onde é necessário armazenamento de informações, porém ele possui a restrição de ser linear. Por exemplo, imagine que queremos armazenar três notas obtidas por quatro alunos diferentes. Neste caso, existe outra estrutura mais adequada para armazenar os dados. A matriz.

A matriz é definida como sendo um vetor com mais de uma dimensão (geralmente duas). Enquanto o vetor armazena as informações de forma linear, a matriz armazena de forma tabular (com linha e colunas).

A imagem a seguir ilustra uma matriz que armazena três notas de quatro alunos:

Posições	**0**	**1**	**2**
0	10	9	6.7
1	6	8	10
2	8	7	4.5
3	5.2	3.3	0.3

Repare que cada linha da matriz representa um aluno que têm três notas (três colunas).

Assim como o vetor, a matriz também possui todos os elementos de um mesmo tipo. Na declaração de uma matriz temos sempre que indicar respectivamente o tipo de dado, nome da variável, número de linhas e colunas (nesta ordem) entre colchetes.

Para fazer acesso a um elemento da matriz, seja para preencher ou para consultar o valor, devemos indicar dois índices, uma para linha e outro para a coluna. O índice é um valor inteiro (pode ser constante ou uma variável) que aparece sempre entre colchetes "[ ]" após o nome do vetor.

Exemplo de Sintaxe

```
01. //Declaração de uma matriz de numeros reais com 5 linhas e 3 colunas
02. real nome_da_variavel[5][3]
03. //Gravar um valor na matriz na posição 0 (primeira linha) e 1 (segunda coluna)
04. nome_da_variavel[0][1] = 2.5
```

Da mesma forma que o vetor, tentar acessar um índice fora do tamanho declarado irá gerar um erro de execução

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         //Declaração de uma matriz de inteiros
06.         // de duas linhas e duas colunas já inicializado.
07.         inteiro matriz[2][2] = {{15,22},{10,11}}
08.
09.         //Atribui -1 na primeira linha e segunda
10.         // coluna da matriz.
11.         matriz[0][1] = -1
12.
13.         //Imprime o valor 15 correspondente
14.         // a primeira linha e primeira coluna da matriz.
15.         inteiro i = 0
16.         escreva(matriz[i][0])
17.         escreva("\n")
18.
```

```
19.         //Imprime o valor 11 correspondente
20.         // a última linha e última coluna da matriz.
21.         escreva(matriz[1][1])
22.
23.         //Declaração de uma matriz de reais de
24.         // duas linhas e quatro colunas.
25.         real outra_matriz[2][4]
26.
27.         //Declaração de uma matriz de caracteres onde o tamanho
28.         // de linhas e colunas são definidos pela inicialização
29.         caracter jogo_velha[][] = {{'X','O','X'}
30.                                   ,{'O','X','O'}
31.                                   ,{' ',' ','X'}}
32.
33.     }
```

Declaração de Vetor

Armazenar a nota de um aluno é possível utilizando uma variável do tipo real. Mas para armazenar as notas de todos os alunos de uma turma? Seria necessário criar uma variável para cada aluno? E se cada turma tiver quantidade de alunos variáveis? E os nomes de cada um dos alunos? Poderíamos armazenar estes dados em variáveis, porém o controle de muitas variáveis em um programa não é uma solução prática. Ao invés disso, utiliza-se uma estrutura de dados que agrupa todos estes valores em um nome único. Esta estrutura chama-se vetor.

Um vetor pode ser visto como uma variável que possui diversas posições, e com isso armazena diversos valores, porém todos do mesmo tipo.

Assim como as variáveis, o vetor tem que ser declarado. Sua declaração é similar à declaração de variáveis, definindo primeiro o seu tipo, em seguida do seu nome e por fim a sua dimensão entre colchetes (opcional se for atribuir valores a ele na declaração).

Exemplo de Sintaxe

```
01. inteiro vetor[5]
02. caracter vetor2[200]
03.
04. //vetores inicializados
05. real vetor3[2] = {1.4,2.5}
06. logico vetor4[4] = {verdadeiro,falso,verdadeiro,verdadeiro}
07. cadeia vetor5[] = {"Questão","Fundamental"}
08.
09. //Mudando o valor do vetor5 na posição 0 de "Questão" para "Pergunta"
10. vetor[0] = "Pergunta"
```

Elementos individuais são acessados por sua posição no vetor. Como um vetor tem mais de uma posição, deve-se indicar qual posição do vetor se quer fazer acesso. Para isso é necessário usarmos um índice.

O índice é um valor inteiro que aparece sempre entre colchetes "[ ]" após o nome do vetor. Adotamos que a primeira posição do vetor tem índice zero (similar a linguagem C) e a última depende do tamanho do vetor. Em um vetor de dez elementos tem-se as posições 0,1,2,3,4,5,6,7,8,9. Já um vetor de quatro elementos tem apenas os índices 0,1,2,3.

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         //Declaração de um vetor de inteiros
06.         // de cinco posições já inicializado.
07.         inteiro vetor[5] = {15,22,8,10,11}
08.
09.         //Imprime o valor 15 correspondente
10.         // ao primeiro elemento do vetor.
11.         escreva(vetor[0])
12.         escreva("\n")
13.
14.         //Imprime o segundo elemento do vetor
15.         escreva(vetor[1])
16.         escreva("\n")
17.
18.         //Imprime o  valor 11 correspondente
19.         // ao último elemento do vetor
```

```
20.         escreva(vetor[4])
21.
22.         //Declaração de um vetor de reais de dez posições
23.         real outro_vetor[10]
24.
25.         //Declaração de um vetor de caracteres onde o tamanho
26.         // é definido pela quantidade de elementos da inicialização
27.         caracter nome[] = {'P','o','r','t','u','g','o','l'}
28.     }
29. }
```

Entrada e Saída

Entrada/saída é um termo utilizado quase que exclusivamente no ramo da computação (ou informática), indicando entrada (inserção) de dados por meio de algum código ou programa, para algum outro programa ou hardware, bem como a sua saída (obtenção de dados) ou retorno de dados, como resultado de alguma operação de algum programa, consequentemente resultado de alguma entrada.

A instrução de entrada de dados possibilita que o algoritmo capture dados provenientes do ambiente externo (fora da máquina) e armazene em variáveis. Assim um algoritmo consegue representar e realizar operações em informações que foram fornecidas por um usuário tais como: nome, idade, salário, sexo, etc. A forma mais comum de capturar dados é através do teclado do computador. Por meio dele o usuário pode digitar números, palavras, frases etc.

A instrução de saída de dados permite ao algoritmo exibir dados na tela do computador. Ela é utilizada para exibir mensagens, resultados de cálculos, informações contidas nas variáveis, etc.

Nesta seção, serão abordados os seguintes tópicos:

- Escreva
- Leia
- Limpa

Escreva

Em determinadas situações precisamos mostrar ao usuário do programa alguma informação. Para isso, existe um comando na programação que exibe dados ao usuário. No portugol a instrução de saída de dados para a tela é chamada de "escreva", pois segue a ideia de que o algoritmo está escrevendo dados na tela do computador.

O comando escreva é utilizado quando deseja-se mostrar informações no console da IDE, ou seja, é um comando de saída de dados.

Para utilizar o comando escreva, você deverá escrever este comando e entre parênteses colocar a(s) variável(eis) ou texto que você quer mostrar no console. Lembrando que quando você utilizar textos, o texto deve estar entre aspas. A sintaxe para utilização deste comando está demonstrada a seguir:

Exemplo de Sintaxe

```
01. escreva ("Escreva o texto a ser digitado aqui")
```

O fluxograma abaixo ilustra as diversas formas de se exibir valores na tela com o comando escreva.



Note que quando queremos exibir o valor de alguma variável não utilizamos as aspas. Para exibição de várias mensagens em sequência, basta separá-las com vírgulas.

Existem duas ferramentas importantes que auxiliam a organização e visualização de textos exibidos na tela. São elas: o quebra-linha e a tabulação.

O quebra-linha é utilizado para inserir uma nova linha aos textos digitados. Sem ele, os textos seriam exibidos um ao lado do outro. Para utilizar este comando, basta inserir "\n". O comando de tabulação é utilizado para inserir espaços maiores entre os textos digitados. Para utilizar este comando, basta inserir "\t".

O exemplo a seguir ilustra em portugol o mesmo algoritmo do fluxograma acima, bem como a utilização do quebra-linha e da tabulação.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         inteiro variavel=5
06.
```

```
07.         //escreve no console um texto qualquer
08.         escreva ("Escreva um texto aqui.\n")
09.
10.         //escreve no console o valor da variável "variavel"
11.         escreva (variavel, "\n")
12.
13.         //escreve no console o resultado da operação
14.         escreva (variavel+variavel, "\n")
15.
16.         //escreve no console o texto digitado, e o valor contido na variável
17.         escreva ("O valor da variável é: ", variavel)
18.
19.         //escreve no console o texto com quebra de linha
20.         escreva ("Texto com\n", "quebra-linha")
21.
```

Leia

Em alguns problemas, precisamos que o usuário digite um valor a ser armazenado. Por exemplo, se quisermos elaborar um algoritmo para calcular a média de nota dos alunos, precisaremos que o usuário informe ao algoritmo quais as suas notas. No portugol a instrução de entrada de dados via teclado é chamada de "leia", pois segue a ideia de que o algoritmo está lendo dados do ambiente externo(usuário) para poder utilizá-los.

O Comando leia é utilizado quando se deseja obter informações do teclado do computador, ou seja, é um comando de entrada de dados. Esse comando aguarda um valor a ser digitado e o atribui diretamente na variável.

Para utilizar o comando leia, você deverá escrever este comando e entre parênteses colocar a(s) variavel (eis) que você quer que recebam os valores a serem digitados. A sintaxe deste comando está exemplificada a seguir:

Exemplo de Sintaxe

```
01. inteiro x
02. cadeia y
03. real z
04.
05. //chamando o comando leia
06. leia(x)
07. leia(y,z)
08. //No final as variáveis irão possuir o valor digitado pelo usuário.
```

Note que para armazenar um valor em uma variável, é necessário que a mesma já tenha sido declarada anteriormente. Assim como no comando escreva, se quisermos que o usuário entre com dados sucessivos, basta separar as variáveis dentro dos parênteses com vírgula.

O fluxograma abaixo ilustra um algoritmo que lê as variáveis: idade, salario, nome, sobrenome, nota1, nota2 e nota3.



O exemplo a seguir ilustra em portugol o mesmo algoritmo do fluxograma acima.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         inteiro idade
06.         real salario, nota1, nota2, nota3
07.         cadeia nome, sobrenome
08.
09.         escreva("Informe a sua idade: ")
10.         leia (idade)                //lê o valor digitado para "idade"
```

```
11.
12.         escreva("Informe seu salario: ")
13.         leia (salario)              //lê o valor digitado para "salario"
14.
15.         escreva("Informe o seu nome e sobrenome: ")
16.         leia (nome, sobrenome)      //lê o valor digitado para "nome" e "sobrenome"
17.
18.         escreva("Informe as suas três notas: ")
19.         leia (nota1, nota2, nota3)  //lê o valor digitado para "nota1", "nota2" e "nota3"
20.
21.         escreva("Seu nome é:"+nome+" "+sobrenome+"\n")
22.         escreva("Você tem "+idade+" anos e ganha de salario "+salario+"\n")
23.         escreva("Suas três notas foram:\n")
24.         escreva("Nota 1: "+nota1+"\n")
25.         escreva("Nota 2: "+nota2+"\n")
```

Limpa

À medida que um algoritmo está sendo executado ele exibe mensagens e executa ações no console. Assim, em alguns casos o console fica poluído com informações desnecessárias, que atrapalham a compreensão e visualização do programa. Para isso, podemos usar o comando limpa.

O comando limpa é responsável por limpar o console. Não requer nenhum parâmetro e não tem nenhuma saída. Sua sintaxe é simples, e está demonstrada a seguir:

Exemplo de Sintaxe

```
01. limpa()
```

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03. 
04.     funcao inicio()
05.     {
06.         cadeia nome
07. 
08.         //imprime a frase "Qual é o seu nome?"
09.         escreva("Qual é o seu nome ?\n")
10. 
11.         //Detecta o que o usuario escreveu na tela
12.         leia(nome)
13. 
14.         //Limpa tudo que estava escrito no console
15.         limpa()
```

Expressões

Uma expressão em uma linguagem de programação é uma combinação de valores explícitos, constantes, variáveis, operadores e funções que são interpretados de acordo com as regras específicas de precedência e de associação para uma linguagem de programação específica, que calcula e, em seguida, produz um outro valor. Este processo, tal como para as expressões matemáticas, chama-se avaliação. O valor pode ser de vários tipos, tais como numérico, cadeia, e lógico.

Nesta seção, serão abordados os seguintes tópicos:

Operações Relacionais

Vamos imaginar que você precise verificar se um número digitado pelo usuário é positivo ou negativo. Como poderíamos verificar isto? Através de uma operação relacional. As operações relacionais também são nossas conhecidas da Matemática. Em algoritmos, os operadores relacionais são importantes, pois permitem realizar comparações que terão como resultado um valor lógico (verdadeiro ou falso).

Os símbolos que usamos para os operadores também mudam um pouco em relação ao que usamos no papel. Os símbolos para diferente, maior ou igual e menor ou igual mudam pois não existem nos teclados convencionais. A tabela a seguir mostra todas as operações relacionais e os símbolos que o Portugol utiliza.

Operação	Símbolo
Maior	>
Menor	<
Maior ou igual	>=
Menor ou igual	<=
Igual	==
Diferente	!=

A tabela a seguir apresenta a estrutura de algumas dessas operações.

Operação	Resultado
3 > 4	Falso
7 != 7	Falso
9 == 10 - 1	Verdadeiro
33 <= 100	Verdadeiro
6 >= 5 + 1	Verdadeiro
5 + 4 <= 11 - 2	Verdadeiro

Nos dois últimos exemplos, temos operadores aritméticos e relacionais juntos. Nestes casos, realiza-se primeiro a operação aritmética e depois a relacional.

Em geral, as operações relacionais são utilizadas em conjunto com as Estruturas de Controle. Veja a sintaxe:

Exemplo de Sintaxe

```
01. se (5 > 3) // Estrutura de controle: "se (...)", Operação relacional: "5 > 3"
02. {
03.     //conjunto de comandos se for verdadeiro
04. }
05.
06. para (i = 0; i < 5; i++) // Estrutura de controle: "para(...)", Operação relacional: "i < 5"
07. {
08.     //conjunto de comandos a serem repetidos até a Expressão se tornar falsa
09. }
10.
11. faca // Estrutura de controle "faca{}enquanto(...)", Operação relacional: "6 < 2"
12. {
13.     //conjunto de comandos a serem repetidos enquanto a condicao for verdadeira após a primeira execuç
14. } enquanto ( 6 < 2 );
```

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
```

```
03.     funcao inicio()
04.     {
05.         //Comparação entre valor A e B utilizando o operador maior que
06.         inteiro a = 5, b = 3
07.         se(a > b){
08.             escreva("A é maior que B")
09.         }
10.
11.         //Comparação entre A e B utilizando o operador igual a
12.         se(a == b){
13.             escreva("A é igual a B")
14.         }
15.
16.         //Comparação entre A e B utilizando o operador maior ou igual a
17.         se(a >= b){
```

Atribuições

Quando criamos uma variável, simplesmente separamos um espaço de memória para um conteúdo. Para especificar esse conteúdo, precisamos de alguma forma determinar um valor para essa variável. Para isso, usamos a operação de atribuição.

A instrução de atribuição serve para alterar o valor de uma variável. Ao fazer isso dizemos que estamos atribuindo um novo valor a esta variável. A atribuição de valores pode ser feita de variadas formas. Pode-se atribuir valores através de constantes, de dados digitados pelo usuário (Leia) ou mesmo através de comparações e operações com outras variáveis já existentes. Neste último caso, após a execução da operação, a variável conterá o valor resultante da operação. O sinal de igual "=" é o símbolo da atribuição no Portugol. A variável a esquerda do sinal de igual recebe o valor das operações que estiverem à direita.

Veja a sintaxe:

Exemplo de Sintaxe

```
01. variavel = 6
02. variavel = variavel2
03. variavel = 6 + 4 / variavel2
04. leia (variavel)
```

Note que uma variável só pode receber atribuições do mesmo tipo que ela. Ou seja, se a variável "b" é do tipo inteiro e a variável "a" é do tipo real, a atribuição não poderá ser realizada.

Existem alguns operandos no Portugol que podem ser utilizados para atribuição de valores. São eles:

Operandos:

```
01. variavel1 += variavel2  // Equivalente a: variavel1 = variavel1 + variavel2;
02. variavel1 -= variavel2  // Equivalente a: variavel1 = variavel1 - variavel2;
03. variavel1 *= variavel2  // Equivalente a: variavel1 = variavel1 * variavel2;
04. variavel1 /= variavel2  // Equivalente a: variavel1 = variavel1 / variavel2;
05. variavel1 %= variavel2  // Equivalente a: variavel1 = variavel1 % variavel2;
06. variavel1 & variavel2   // Equivalente a: variavel1 = variavel1 & variavel2;
07. variavel1 ^= variavel2  // Equivalente a: variavel1 = variavel1 ^ variavel2;
08. variavel1 |= variavel2  // Equivalente a: variavel1 = variavel1 | variavel2;
09. variavel1++             // Equivalente a: variavel1 = variavel1 + 1;
10. variavel1--             // Equivalente a: variavel1 = variavel1 - 1;
```

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         //Atribuição de valores constantes a uma variável
06.         inteiro a
07.         a = 2
08.
09.         //Atribuição através de entrada de dados, informado pelo usuário
10.         inteiro b
11.         leia(b)
12.
13.         //Atribuição através de uma variável já informada pelo usuário
14.         inteiro c
15.         c = b
16.     }
```

```
17. }
```

Operações Aritméticas

As operações aritméticas são nossas velhas conhecidas da Matemática. Em algoritmos é muito comum usarmos operadores aritméticos para realizar cálculos.

Os símbolos que usamos para os operadores na Matemática mudam um pouquinho em algoritmos. A multiplicação, que na matemática é um xis 'x' ou um ponto "." torna-se um '*', justamente para não confundir com o xis que pode ser uma variável e com o ponto que pode ser a parte decimal de um número real. A tabela a seguir mostra quais são os operadores que o Portugol utiliza:

Operação	Símbolo	Prioridade
Adição	+	1
Subtração	-	1
Multiplicação	*	2
Divisão	/	2
Resto da divisão inteira	%	2

A prioridade indica qual operação deve ser realizada primeiro quando houverem várias juntas. Quanto maior a prioridade, antes a operação ocorre. Por exemplo:

6 + 7 * 9

A multiplicação 7 * 9 é feita antes pois a operação de multiplicação tem prioridade maior que a soma. O resultado deste cálculo será 69.

O uso de parênteses permite modificar a ordem em que as operações são realizadas. Na Matemática existem os parênteses '(', os colchetes '[' e as chaves '{' para indicar as prioridades. Na computação, usa-se somente os parênteses, sendo que os mais internos serão realizados primeiro.

Nesta seção, serão abordados os seguintes tópicos:

Operação de Adição

Adição é uma das operações básicas da álgebra. Na sua forma mais simples, adição combina dois números (termos, somandos ou parcelas), em um único número, a soma ou total. Adicionar mais números corresponde a repetir a operação.

A sintaxe é bem fácil, se coloca os operandos entre o sinal de mais.

Exemplo de Sintaxe

```
01. escreva(1 + 5) //Operação Aritmética 1 + 5 sendo escrita na tela
02.
03. real numero = 50 + 30 //Operação Aritmética 50 + 30 sendo armazenada na variável  numero
04.
05. se(20 + 40 < 70) //Operação Aritmética 20 + 40 dentro de uma estrutura de controle "se" em conjunto co
06. {
07.     //Comandos
08. }
```

Note que você poderá atribuir o resultado desta operação a uma variável, ou mesmo executar diretamente através do comando escreva.

Propriedades importantes

- **Comutatividade** A ordem das parcelas não altera o resultado da operação. Assim, se 2 + 3 = 5, logo 3 + 2 = 5.
- **Associatividade** O agrupamento das parcelas não altera o resultado. Assim, se (2 + 3) + 1 = 6, logo 2 + (3 + 1) = 6.
- **Elemento neutro** A parcela 0 (zero) não altera o resultado das demais parcelas. O zero é chamado "elemento neutro" da adição. Assim, se 2 + 3 = 5, logo 2 + 3 + 0 = 5.
- **Fechamento** A soma de dois números reais será sempre um número do conjunto dos números reais.
- **Anulação** A soma de qualquer número e o seu oposto é zero. Exemplo:
  - 2 + (-2) = 0
  - (-999) + 999 = 0

Tabela de compatibilidade de tipos da operação de adição

Operando Esquerdo	Operando Direito	Tipo Resultado	Exemplo	Resultado
cadeia	**cadeia**	**cadeia**	"Oi" + " mundo"	"Oi mundo"
cadeia	**caracter**	**cadeia**	"Banan" + 'a'	"Banana"
cadeia	**inteiro**	**cadeia**	"Faz um" + 21	"Faz um 21"
cadeia	**real**	**cadeia**	"Altura: " + 1.78	"Altura: 1.78"
cadeia	**logico**	**cadeia**	"Help bom =" + verdadeiro	"Help bom = verdadeiro"
caracter	**cadeia**	**cadeia**	'P' + "anqueca"	"Panqueca"
caracter	**caracter**	**cadeia**	'C' + 'a' + 'd' + 'e' + 'i' + 'a'	"Cadeia"
inteiro	**cadeia**	**cadeia**	22 + " de agosto"	"22 de agosto"
inteiro	**inteiro**	**inteiro**	12 + 34	46
inteiro	**real**	**real**	76 + 3.25	79.25
real	**cadeia**	**cadeia**	3.24 + " Kg"	"3.24 Kg"
real	**inteiro**	**real**	9.87 + 1	10.87
real	**real**	**real**	9.87 + 0.13	10.0
logico	**cadeia**	**cadeia**	verdadeiro + " amigo"	"verdadeiro amigo"

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo de Sintaxe

```
01. programa
```

```
02. {
03.     funcao inicio()
04.     {
05.     inteiro valor
06.
07.     escreva (5+8, "\n")
08.
09.     valor = 5+8
10.
11.     escreva (valor)
12.     }
13. }
```

Operação de Subtração

Subtração é uma operação matemática que indica quanto é um valor numérico (minuendo) se dele for removido outro valor numérico (subtraendo).

A subtração é o mesmo que a adição por um número de sinal inverso. É, portanto, a operação inversa da adição. Seus elementos estão demonstrados na figura a seguir:



Exemplo de Sintaxe

```
01. escreva(1 - 5) //Operação Aritmética 1 - 5 sendo escrita na tela
02.
03. real numero = 50 - 30 //Operação Aritmética 50 - 30 sendo armazenada na variável numero
```

Note que você poderá atribuir o resultado desta operação a uma variável, ou mesmo executar diretamente através do comando escreva.

Propriedades importantes

- **Fechamento** A diferença de dois números reais será sempre um número real.
- **Elemento neutro** Na subtração não existe um elemento neutro **n** tal que, qualquer que seja o real "a", a - n = n - a = a.
- **Anulação** Quando o minuendo é igual ao subtraendo, a diferença será 0 (zero).

Tabela de compatibilidade de tipos da operação de subtração

Operando Esquerdo	Operando Direito	Tipo Resultado	Exemplo	Resultado
inteiro	**inteiro**	**inteiro**	20 - 10	10
inteiro	**real**	**real**	90 - 0.5	89.5
real	**inteiro**	**real**	11.421 - 3	8.421
real	**real**	**real**	12.59 - 24.59	-12.0

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.     inteiro valor
06.
07.     escreva (10-3, "\n")
08.
09.     valor = 10-3
10.
11.     escreva (valor)
12.     }
13. }
```

Operação de Multiplicação

Na sua forma mais simples a multiplicação é uma forma de se adicionar uma quantidade finita de números iguais. O resultado da multiplicação de dois números é chamado produto. Os números sendo multiplicados são chamados de coeficientes ou operandos, e individualmente de multiplicando e multiplicador, conforme figura abaixo:



Exemplo de Sintaxe

```
01. escreva(1 * 5) //Operação Aritmética 1 * 5 sendo escrita na tela
02.
03. real numero = 50 * 30 //Operação Aritmética 50 * 30 sendo armazenada na variável numero
```

Note que você poderá atribuir o resultado desta operação a uma variável, ou mesmo executar diretamente através do comando escreva.

Propriedades importantes

- **Comutatividade** A ordem dos fatores não altera o resultado da operação. Assim, se x * y = z, logo y * x = z.
- **Associatividade** O agrupamento dos fatores não altera o resultado.(Podemos juntar de dois em dois de modo que facilite o cálculo). Assim, se (x * y) * z = w, logo x * (y * z) = w.
- **Distributividade** Um fator colocado em evidência numa soma dará como produto a soma do produto daquele fator com os demais fatores. Assim, x * (y + z) = (x * y) + (x * z).
- **Elemento neutro** O fator 1 (um) não altera o resultado dos demais fatores. O um é chamado "Elemento neutro" da multiplicação. Assim, se x * y = z, logo x * y * 1 = z.(obs:o 0 é o da soma.)
- **Elemento opositor** O fator -1 (menos um) transforma o produto em seu simétrico. Assim, -1 * x = -x e -1 * y = -y, para y diferente de x.
- **Fechamento** O produto de dois números reais será sempre um número do conjunto dos números reais.
- **Anulação** O fator 0 (zero) anula o produto. Assim, x * 0 = 0, e y * 0 = 0, com x diferente de y.

Tabela de compatibilidade de tipos da operação de multiplicação

Operando Esquerdo	Operando Direito	Tipo Resultado	Exemplo	Resultado
inteiro	**inteiro**	**inteiro**	6 * 8	48
inteiro	**real**	**real**	4 * 1.11	4.44
real	**inteiro**	**real**	6.712 * 174	1167.888
real	**real**	**real**	207.65 * 1.23	255.4095

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.     inteiro valor
06.
07.     escreva (3*4, "\n")
08.
09.     valor = 3*4
10.
11.     escreva (valor)
```

```
12.         }
13.     }
```

Operação de Divisão

Divisão é a operação matemática inversa da multiplicação. É utilizada para, como o próprio nome sugere, dividir, repartir, separar algum valor em partes iguais. Seus elementos estão demonstrados na figura a seguir:



Exemplo de Sintaxe

```
01. escreva(15 / 5) //Operação Aritmética 1 * 5 sendo escrita na tela
02.
03. real numero = 50 / 25.6 //Operação Aritmética 50 * 30 sendo armazenada na variável numero
```

Note que você poderá atribuir o resultado desta operação a uma variável, ou mesmo executar diretamente através do comando escreva.

Propriedades importantes

- **Reintegradora**

  Multiplicando o quociente pelo divisor se obtém o dividendo. Assim, 2 * 10 = 20.

- **Associatividade**

  Quando o divisor for 1, o dividendo e o quociente são iguais. Assim, 20 / 1 = 20.

Tabela de compatibilidade de tipos da operação de divisão

Operando Esquerdo	Operando Direito	Tipo Resultado	Exemplo	Resultado
inteiro	**inteiro**	**inteiro**	5 / 2	2
inteiro	**real**	**real**	125 / 4.5	27.777777
real	**inteiro**	**real**	785.4 / 3	261.8
real	**real**	**real**	40.351 / 3.12	12.9333333

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.     inteiro valor
06.
07.     escreva (20/10, "\n")
08.
09.     valor = 20/10
10.
11.     escreva (valor)
12.     }
```

```
13. }
```

Operação de Módulo

Em algumas situações faz-se necessário manipular o resto de algumas divisões. Por exemplo, se você quiser saber se um determinado valor é par ou ímpar, como faria? Para isso podemos utilizar o módulo. A operação módulo encontra o resto da divisão de um número por outro.

Dados dois números a (o dividendo) e b o divisor, a modulo b (a % b) é o resto da divisão de a por b. Por exemplo, 7 % 3 seria 1, enquanto 9 % 3 seria 0.

Exemplo de Sintaxe

```
01. escreva(13 % 5) //Operação Aritmética 13 % 5 sendo escrita na tela
02.
03. real numero = 50 % 4 //Operação Aritmética 50 % 4 sendo armazenada na variável numero
```

Note que você poderá atribuir o resultado desta operação a uma variável, ou mesmo executar diretamente através do comando escreva.

Tabela de compatibilidade de tipos da operação de módulo

Operando Esquerdo	Operando Direito	Tipo Resultado	Exemplo	Resultado
inteiro	**inteiro**	**inteiro**	45 % 7	3

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.     inteiro valor
06.
07.     escreva (7%3, "\n")
08.
09.     valor = 7%3
10.
11.     escreva (valor)
12.     }
13. }
```

Operações Lógicas

As operações lógicas são uma novidade para muitos, pois raramente são vistas na escola. Um operador lógico opera somente valores lógicos, ou seja, é necessário que o valor à esquerda e a direita do operador sejam valores lógicos (verdadeiro ou falso).

É muito comum usar expressões relacionais (que dão resultado lógico) e combiná-las usando operadores lógicos. Por exemplo:

Operações	Resultado
5 > 3 **e** 2 < 1	falso
nao (8 < 4)	verdadeiro
1 > 3 **ou** 1 <= 1	verdadeiro

Assim como as operações aritméticas, as operações lógicas também possuem prioridades. Veja a tabela abaixo:

Operador	Prioridade
ou	1
e	2
nao	3

Ou seja, o **nao** tem maior prioridade que todos, e o **ou** tem a menor. Veja os exemplos a seguir:

Passo	Exemplo 1	Exemplo 2
Passo 1	**nao** verdadeiro **ou** falso	verdadeiro **e** falso **ou** verdadeiro
Passo 2	falso **ou** falso	falso **ou** verdadeiro
Passo 3	falso	verdadeiro

Nesta seção, serão abordados os seguintes tópicos:

- e
- ou
- nao

e

Em algumas situações, necessitamos que alguma instrução só seja executada se outras condições forem verdadeiras. Por exemplo, se você quisesse testar se duas variáveis distintas têm valor igual a 2, como faria? Para isso podemos utilizar o operador lógico **e**.

Quando usamos o operador **e** o resultado de uma operação lógica será verdadeiro somente quando AMBOS os operandos forem verdadeiros. Ou seja, basta que um deles seja falso e a resposta será falsa. A tabela a seguir é conhecida como tabela verdade e ilustra o comportamento do operador **e**.

Operação 1	Operação 2	Operação 1 e Operação 2
Verdadeiro	Verdadeiro	Verdadeiro
Verdadeiro	Falso	Falso
Falso	Verdadeiro	Falso
Falso	Falso	Falso

Em geral, os operadores lógicos são utilizados em conjunto com as Estruturas de Controle.

Exemplo de Sintaxe

```
01. se (5 > 4 e 6 == 6) //Operação logica 'e' junto com operações relacionais.
02. {
03.     //comandos
04. }
05. 
06. enquanto(verdadeiro e 5 < 4) //Operação logica 'e' junto com operações relacionais e tipo logico.
07. {
08.     //comandos
09. }
10. 
11. logico saida = 5 > 3 e 4 < 5 e 6 < 7 //Operação lógica 'e' junto com operações relacionais.
```

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         //Teste utilizando o operador lógico "e" onde a deve ser igual a 2 e b deve ser igual a 2 tamb
06.         inteiro a = 2, b = 2
07.         se(a == 2 e b == 2)
08.         {
09.             escreva("Teste positivo")
10.         }
11. 
12.         //Neste caso c é igual a 2, entretanto d não é igual a 2, logo este teste não terá como respos
13.         inteiro c = 2, d = 3
14.         se(c == 2 e d == 2)
15.         {
16.             escreva("Teste positivo")
17.         }
18. 
19.         //Neste caso de teste g é igual a 2 e f é diferente de 3, logo este teste terá como resposta v
20.         inteiro g = 2, f = 2
21.         se(g == 2 e f != 3)
22.         {
23.             escreva("Teste positivo")
```

```
24.             }
25.         }
26.     }
```

ou

Em algumas situações, necessitamos que alguma instrução seja executada se uma entre várias condições forem verdadeiras. Por exemplo, se você quisesse testar se pelo menos uma entre duas variáveis distintas têm valor igual a 2, como faria? Para isso podemos utilizar o operador lógico **ou**.

Quando usamos o operador **ou** o resultado de uma operação lógica será verdadeiro sempre que UM dos operandos for verdadeiro. A tabela verdade a seguir ilustra o comportamento do operador **ou**.

Operação 1	Operação 2	Operação 1 **ou** Operação 2
Verdadeiro	Verdadeiro	Verdadeiro
Verdadeiro	Falso	Verdadeiro
Falso	Verdadeiro	Verdadeiro
Falso	Falso	Falso

Em geral, os operadores lógicos são utilizados em conjunto com as Estruturas de Controle.

Exemplo de Sintaxe

```
01. se (5 > 4 ou 7 == 6) //Operação logica 'ou' junto com operações relacionais.
02. {
03.     //comandos
04. }
05. 
06. enquanto(falso ou 5 > 4) //Operação logica 'ou' junto com operações relacionais e tipo logico.
07. {
08.     //comandos
09. }
10. 
11. logico saida = 5 > 8 ou 4 < 12 ou 34 < 7 //Operação lógica 'ou' junto com operações relacionais.
```

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         //Teste utilizando o operador lógico "ou" onde a deve ser igual a 2 ou pelo menos b deve ser i
06.         inteiro a = 2, b = 2
07.         se(a == 2 ou b == 2)
08.         {
09.             escreva("Teste positivo")
10.         }
11. 
12.         //Neste caso c é igual a 2, entretanto d não é igual a 2, mas qualquer uma das condições ofere
13.         inteiro c = 2, d = 3
14.         se(c == 2 ou d == 2)
15.         {
```

nao

Em algumas situações necessitamos verificar se o contrário de uma sentença é verdadeiro ou não. Por exemplo, se você tem uma variável com um valor falso, e quer fazer um teste que será positivo sempre que essa variável for falsa, como faria? Para isso podemos utilizar o operador lógico **nao**.

O operador **nao** funciona de forma diferente pois necessita apenas de um operando. Quando usamos o operador **nao**, o valor lógico do operando é invertido, ou seja, o valor falso torna-se verdadeiro e o verdadeiro torna-se falso.

Em geral, os operadores lógicos são utilizados em conjunto com as Estruturas de Controle.

Exemplo de Sintaxe

```
01. se (nao falso) //Operação logica 'nao' junto com operações relacionais.
02. {
03.     //comandos
04. }
05. 
06. enquanto(nao 5 < 4) //Operação logica 'nao' junto com operações relacionais e tipo logico.
07. {
08.     //comandos
09. }
10. 
11. logico saida = nao (5 > 3 e 4 < 5) e 6 < 7 //Operação lógica 'nao' junto com operações relacionais e o
```

Para melhor compreensão deste conceito, confira o exemplo abaixo.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         //Neste caso de teste a variável teste foi inicializada como falso, e foi verificado se teste
06.         logico teste = falso
07.         se(nao(teste))
08.         {
09.             escreva("Teste positivo")
10.         }
11. 
12.         //Neste caso teste a soma das variáveis a e b resulta em 5, e comparado se a mesma é maior que
13.         inteiro a = 2, b = 3
14.         se(nao(a+b > 7))
15.         {
```

Operações Bitwise

Operadores bitwise são utilizados quando precisamos realizar operações em termos de bits, porém utilizando valores inteiros, ou seja, trabalhar com sua representação binária. São semelhantes aos operadores lógicos, porém trabalham com representação de dados binária.

Este tipo de operador analisa cada bit dos valores individualmente, realizando a instrução de acordo com o operando utilizado. Cada comparação que for feita, ou seja, cada bit analisado poderá ser considerado como uma variável lógica, assumindo valores de verdadeiro (1) ou falso (0).

Complemento de Dois

O Portugol utiliza complemento de dois para representar os inteiros negativos em base binária. Em computação, complemento para dois ou complemento de dois é um tipo de representação binária de números com sinal amplamente usada nas arquiteturas dos dispositivos computacionais modernos.

O dígito mais significativo (MSB) é o que informa o sinal do número. É o dígito localizado mais a esquerda do número. Se este dígito for 0 o número é positivo, e se for 1 é negativo.

Os números são escritos da seguinte forma:

- Positivos: Sua magnitude é representada na sua forma binária direta, e um bit de sinal 0 é colocado na frente do MSB.
  - (bit 0) + o número em binário.
  - Exemplos: 0001 (+1), 0100 (+4) e 0111 (+7)
- Negativos: Sua magnitude é representada na forma de complemento de 2, e um bit de sinal é colocado na frente do MSB.
  - Pegamos o número em binário e "invertemos" (0100 invertendo têm-se 1011) e somamos um ao valor "invertido" (1011 + 0001 = 1100).

As vantagens do uso do complemento de 2 são: existe somente um zero e as regras para soma e subtração são as mesmas. A desvantagem é o fato de ser um código assimétrico.

Tabela exemplo do complemento de dois para um binário inteiro de 4 bits

Complemento de dois	Decimal
0111	7
0110	6
0101	5
0100	4
0011	3
0010	2
0001	1
0000	0
1111	-1
1110	-2
1101	-3
1100	-4
1011	-5
1010	-6
1001	-7
1000	-8

Nesta seção, serão abordados os seguintes tópicos:

- Operação de Bitwise AND (&)

- Operação de Bitwise NOT (~)
- Operação de Bitwise OR (|)
- Operação de Bitwise Shift (<<) ou (>>)
- Operação de Bitwise XOR (^)

Operação de Bitwise AND

Muito semelhante ao operador lógico 'e', o operador binário AND, ou conjunção binária devolve um bit 1 sempre que ambos operandos forem '1', conforme podemos confirmar pela tabela verdade, onde A e B são bits de entrada e S é o bit-resposta, ou bit de saída:

B A S
0 0 0
0 1 0
1 0 0
1 1 1

Sua sintaxe é o operador '&' entre os dois inteiros.

Exemplo de Sintaxe

```
01. /*
02. Para se fazer a operação Bitwise 5 AND 3
03. 0101 (decimal 5)
04. AND
05. 0011 (decimal 3)
06. ----
07. 0001 (decimal 1)
08. */
09. inteiro resultado = 5 & 3
```

Tabela de compatibilidade de tipos da operação de Bitwise AND

Operando Esquerdo	Operando Direito	Tipo Resultado	Exemplo	Resultado
inteiro	**inteiro**	**inteiro**	6 & 13	4

Lembre-se que os operadores bitwise só trabalham com números do Tipo Inteiro. O exemplo a seguir ilustra em portugol o mesmo exemplo usado anteriormente.

Exemplo de Sintaxe

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         escreva (5 & 3)
06.     }
07. }
```

Operação de Bitwise OR

O operador binário OR, ou disjunção binária devolve um bit 1 sempre que pelo menos um dos operandos seja '1', conforme podemos confirmar pela tabela de verdade, onde A e B são os bits de entrada e S é o bit-resposta, ou bit de saída:

B A S
0 0 0
0 1 1
1 0 1
1 1 1

Sua sintaxe é o operador '|' (Digito de Canalização (em inglês: pipe)) entre os dois inteiros.

Exemplo de Sintaxe

```
01. /*
02. Para se fazer a operação Bitwise 5 OR 3
03. 0101 (decimal 5)
04. OR
05. 0011 (decimal 3)
06. ----
07. 0111 (decimal 7)
08. */
09. inteiro resultado = 5 | 3
```

Tabela de compatibilidade de tipos da operação de Bitwise OR

Operando Esquerdo	Operando Direito	Tipo Resultado	Exemplo	Resultado
inteiro	**inteiro**	**inteiro**	2 | 8	10

Lembre-se que os operadores bitwise só trabalham com números do Tipo Inteiro. O exemplo a seguir ilustra em portugol o mesmo exemplo usado anteriormente.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         escreva (5 | 3)
06.     }
07. }
```

Operação de Bitwise NOT

Muito semelhante ao operador lógico 'nao', o operador unário NOT, ou negação binária devolve um bit 1 sempre que ambos operandos forem '1', conforme podemos confirmar pela tabela de verdade, onde A é o bit de entrada e S é o bit-resposta, ou bit de saída:

A S
0 1
1 0

Sua sintaxe é o operador '~' entre os dois inteiros.

Exemplo de Sintaxe

```
01. /*
02. Para se fazer a operação Bitwise NOT 7
03. 0111  (decimal 7)
04. NOT
05. ----
06. 1000  (decimal 8)
07. */
08. inteiro resultado = ~7
```

Tabela de compatibilidade de tipos da operação de Bitwise NOT

Operando Tipo	Resultado	Exemplo	Resultado
inteiro	**inteiro**	~ 1	-2

Lembre-se que os operadores bitwise só trabalham com números do Tipo Inteiro. O exemplo a seguir ilustra em portugol o mesmo exemplo usado anteriormente. É importante a compreensão do conceito "Complemento de dois" presente no menu "Operações Bitwise".

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         escreva (~7)
06.     }
07. }
```

Operação de Bitwise XOR

O operador binário XOR, ou disjunção binária exclusiva devolve um bit '1' sempre que o número de operandos iguais a 1 é ímpar, conforme podemos confirmar pela tabela de verdade, onde A e B são bits de entrada e S é o bit-resposta, ou bit de saída:

B A S
0 0 0
0 1 1
1 0 1
1 1 0

Sua sintaxe é o operador '^' entre os dois inteiros

Exemplo de Sintaxe

```
01. /*
02. Para se fazer a operação Bitwise 5 XOR 3
03. 0101 (decimal 5)
04. XOR
05. 0011 (decimal 3)
06. ----
07. 0110 (decimal 6)
08. */
09. inteiro resultado = 5 ^ 3
```

Tabela de compatibilidade de tipos da operação de Bitwise XOR

Operando Esquerdo Operando Direito Tipo Resultado Exemplo Resultado
inteiro **inteiro** **inteiro** 2 ^ 10 8

Lembre-se que os operadores bitwise só trabalham com números do Tipo Inteiro. O exemplo a seguir ilustra em portugol o mesmo exemplo usado anteriormente.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         escreva (5 ^ 3)
06.     }
07. }
```

Operação de Bitwise Shift

Os operadores de Bitwise Shift são utilizados para deslocar bits de um número inteiro para direita ou para a esquerda.

Left Shift

Em um deslocamento aritmético à esquerda, os bits são deslocados para a esquerda e zeros são acrescentados à direita como demonstra a imagem:



Sua sintaxe respectivamente o valor inteiro, o operador '<<' e o numero de bits a ser deslocado

Exemplo de Sintaxe

```
01. /*
02. Para se fazer a operação Bitwise 23 >> 1
03. 00010111 (decimal +23)
04. LEFT-SHIFT uma vez
05. --------
06. 00101110 (decimal +46)
07. */
08. inteiro resultado = 23 << 1
```

O número de bits a ser deslocado equivale a quantidade de vezes que o valor será multiplicado por 2.

Right Shift

Em um deslocamento aritmético para a direita, o bit de sinal é deslocado da esquerda, preservando, assim, o sinal do operando como demonstra a imagem:



Sua sintaxe respectivamente o valor inteiro, o operador '>>' e o numero de bits a ser deslocado

Exemplo de Sintaxe

```
01. /*
02. Para se fazer a operação Bitwise -105 >> 1
03. 10010111 (decimal -105)
04. RIGHT-SHIFT uma vez
05. --------
```

```
11001011 (decimal -53)
*/
inteiro resultado = -105 >> 1
```

O número de bits a ser deslocado equivale a quantidade de vezes que o valor será dividido por 2, sempre resultando em um valor inteiro.

Tabela de compatibilidade de tipos da operação de Bitwise SHIFT

Operando Esquerdo	Operando Direito	Tipo Resultado	Exemplo Resultado
inteiro	**inteiro**	**inteiro**	12 >> 2 3
inteiro	**inteiro**	**inteiro**	12 << 2 48

O exemplo a seguir ilustra em portugol os mesmos exemplos usados anteriormente.

Exemplo

```
programa
{
    funcao inicio()
    {
        escreva (23 << 1, "\n", -105 >> 1)
    }
}
```

Estruturas de Controle

Em determinadas situações é necessário executar ações de acordo com os dados fornecidos pelo programa. Em alguns casos, pode ser necessário que o programa execute uma determinada instrução repetidas vezes, por exemplo. Sendo assim, controlar e manipular a ordem com que instruções serão executadas em função de dados fornecidos pelo programa é essencial, e é para isso que servem as estruturas de controle.

Estruturas de controle (ou fluxos de controle) referem-se à ordem em que instruções, expressões e chamadas de função são executadas. Sem o uso de estruturas de controle, o programa seria executado de cima para baixo, instrução por instrução, dificultando assim a resolução de diversos problemas.

Nesta seção, serão abordados os seguintes tópicos:

Desvios Condicionais

Não é só na vida que fazemos escolhas. Nos algoritmos encontramos situações onde um conjunto de instruções deve ser executado caso uma condição seja verdadeira. Por exemplo: sua aprovação na disciplina de algoritmos depende da sua média final ser igual ou superior a 6. Podemos ainda pensar em outra situação: a seleção brasileira de futebol só participa de uma copa do mundo se for classificada nas eliminatórias, se isso não ocorrer ficaremos sem o hexacampeonato.

Estas e outras situações podem ser representadas nos algoritmos por meio de desvios condicionais.

Nesta seção, serão abordados os seguintes tópicos:

- se
- se-senao
- se-senao-se
- escolha-caso

Se

Aqui veremos como dizer a um algoritmo quando um conjunto de instruções deve ser executado. Esta determinação é estabelecida se uma condição for verdadeira. Mas o que seria esta condição? Ao executar um teste lógico teremos como resultado um valor verdadeiro ou falso. A condição descrita anteriormente nada mais é que um teste lógico.

Se este teste lógico resultar verdadeiro, as instruções definidas dentro do desvio condicional serão executadas. Se o teste for falso, o algoritmo pulará o trecho e continuará sua execução a partir do ponto onde o desvio condicional foi finalizado.

O desvio condicional que foi acima apresentado é considerado simples e conhecido como o comando **se**.

A sintaxe é respectivamente a palavra reservada **se**, a condição a ser testada entre parenteses e as instruções que devem ser executadas entre chaves caso o desvio seja verdadeiro.

Exemplo de Sintaxe

```
01. logico condicao = verdadeiro
02. se (condicao)
03. {
04.     //Instruções a serem executadas se o desvio for verdadeiro
05. }
06.
07. inteiro x = 5
08. se (x > 3)
09. {
10.     //Instruções a serem executadas se o desvio for verdadeiro
11. }
```

A figura abaixo ilustra um algoritmo que verifica se o número digitado pelo usuário é zero. Ele faz isso usando um desvio condicional. Note que se o teste for verdadeiro exibirá uma mensagem, no caso falso nenhuma ação é realizada.



O exemplo a seguir ilustra em portugol o mesmo algoritmo do fluxograma acima.

Exemplo

```
01. programa
02. {
03.         funcao inicio()
04.         {
05.
06.             inteiro num
07.
08.             escreva ("Digite um número: ")
09.             leia (num)
10.
11.             se (num==0)
12.             {
13.                 escreva ("O número digitado é 0")
14.             }
15.
```

Se-senao

Agora vamos imaginar que se a condição for falsa um outro conjunto de comandos deve ser executado. Quando iremos encontrar esta situação?

Imagine um programa onde um aluno com média final igual ou maior a 6 é aprovado. Se quisermos construir um algoritmo onde após calculada a média, seja mostrada na tela uma mensagem indicando se o aluno foi aprovado ou reprovado. Como fazer isto? Utilizando o comando **se** junto com o **senao**.

Sua sintaxe é simples, basta no termino do comando se ao lado do fechamento de chaves, colocar o comando **senao** e entre chaves colocar as instruções a serem executadas caso o comando **se** for falso

Exemplo de Sintaxe

```
01. logico condicao = falso
02. se (condicao)
03. {
04.     //Instruções a serem executadas se o desvio for verdadeiro
05. }
06. senao
07. {
08.     //Instruções a serem executadas se o desvio for falso
09. }
```



O exemplo a seguir ilustra em portugol o mesmo algoritmo do fluxograma acima.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.
06.         inteiro hora
07.
08.         escreva ("Digite a hora: ")
09.         leia (hora)
10.
```

```
11.         se (hora >= 6 e hora <= 18)
12.         {
13.             escreva ("É dia")
14.         }
15.         senao
16.         {
17.             escreva ("É noite")
18.         }
19.
20.     }
21. }
```

Se-senao se

Agora imagine que você precise verificar a nota da prova de um aluno e falar se ele foi muito bem, bem, razoável ou mau em uma prova como fazer isto ?

Quando você precisa verificar se uma condição é verdadeira, e se não for, precise verificar se outra condição é verdadeira uma das formas de se fazer esta verificação é utilizando do **se** ... **senao se**;

A sua sintaxe é parecida com a do **senao**, mas usando o comando **se** imediatamente após escrever o comando **senao**.

Exemplo de Sintaxe

```
01. logico condicao = falso
02. logico condicao2 = verdadeiro
03. se (condicao)
04. {
05.     //Instruções a serem executadas se o desvio for verdadeiro
06. }
07. senao se (condicao2)
08. {
09.     //Instruções a serem executadas se o desvio anterior for falso e este desvio for verdadeiro
10. }
```

Também pode-se colocar o comando **senao** no final do ultimo **senao se**, assim quando todos os testes falharem, ele irá executar as instruções dentro do **senao**

Exemplo de Sintaxe

```
01. se (12 < 5)
02. {
03.     //Instruções a serem executadas se o desvio for verdadeiro
04. }
05. senao se ("palavra" == "texto")
06. {
07.     //Instruções a serem executadas se o desvio anterior for falso e este desvio for verdadeiro
08. }
09. senao
10. {
11.     //Instruções a serem executadas se o desvio anterior for falso
12. }
```

O exemplo a seguir ilustra a resolução do em Portugol de avisar se o aluno foi muito bem, bem, razoável ou mau em uma prova.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         real nota
06.         leia(nota)
07.         se(nota >= 9)
08.         {
09.             escreva("O aluno foi um desempenho muito bom na prova")
10.         }
11.         senao se (nota >= 7)
12.         {
13.             escreva("O aluno teve um desempenho bom na prova")
```

```
14.         }
15.         senao se (nota >= 6)
16.         {
17.             escreva("O aluno teve um desempenho razoável na prova")
18.         }
19.         senao
20.         {
21.             escreva("O aluno teve um desempenho mau na prova")
22.         }
23.     }
24. }
```

Escolha-caso

Qual a melhor forma para programar um menu de, por exemplo, uma calculadora? Esta tarefa poderia ser executada através de desvios condicionais **se** e **senão**, porém esta solução seria complexa e demorada. Pode-se executar esta tarefa de uma maneira melhor, através de outro tipo de desvio condicional: o **escolha** junto com o **caso**. Este comando é similar aos comandos **se** e **senão**, e reduz a complexidade do problema.

Apesar de suas similaridades com o **se**, ele possui algumas diferenças. Neste comando não é possível o uso de operadores lógicos, ele apenas trabalha com valores definidos, ou o valor é igual ou diferente. Além disto, o **escolha** e o **caso** tem alguns casos testes, e se a instrução **pare** não for colocada ao fim de cada um destes testes, o comando executará todos casos existentes.

A sintaxe do **escolha** é respectivamente o comando **escolha** a condição a ser testada e entre chaves se coloca os casos

A sintaxe para se criar um caso é a palavra reservada **caso**, o valor que a condição testada deve possuir, dois pontos e suas instruções. Lembre-se de termina-las com o comando **pare**

Exemplo de Sintaxe

```
01. inteiro numero
02. escolha(numero)
03. {
04.     caso 1:
05.         //Instruções caso o numero for igual a 1
06.     pare
07.
08.     caso 2:
09.         //Instruções caso o numero for igual a 2
10.     pare
11.
12.     caso 50:
13.         //Instruções caso o numero for igual a 50
14.     pare
15.
```

O comando pare evita que os blocos de comando seguinte sejam executados por engano. O caso contrario será executado caso nenhuma das expressões anteriores sejam executadas.

A figura a seguir ilustra um algoritmo que verifica se o a variável valor é igual a 0, 1 ou 2.



O exemplo a seguir ilustra em portugol o mesmo algoritmo do fluxograma acima.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.     inteiro valor=1
06.         escolha (valor)
07.         {
08.         caso 0:     //testa se o valor é igual a 0
09.         escreva ("o valor é igual a 0")
10.         pare
11.
12.         caso 1:     //testa se o valor é igual a 1
13.         escreva ("o valor é igual a 1")
14.         pare
15.
16.         caso 2:     //testa se o valor é igual a 2
17.         escreva ("o valor é igual a 2")
18.         pare
19.
```

```
20.         caso contrario:
21.         escreva ("o valor não é igual a 0, 1 ou 2")
22.         }
23.     }
24. }
```

Laços de Repetição

Existem problemas que são repetitivos por natureza. Por exemplo, escrever um algoritmo para calcular a média de um aluno é algo fácil, mas se quisermos calcular a média da turma inteira? A solução mais simples seria executar o algoritmo tantas vezes que o cálculo fosse necessário, embora esta tarefa seja um tanto trabalhosa. Mas se ainda, nesta conta ao final o professor quisesse que fosse mostrada a média mais alta e mais baixa da turma?

Esse e outros problemas podem ser resolvidos com a utilização de laços de repetição. Um laço de repetição, como sugere o próprio nome, é um comando onde uma quantidades de comandos se repete até que uma determinada condição seja verdadeira.

O Portugol contém 3 tipos de laços de repetição: pré-testado, pós-testado e laço com variável de controle.

Nesta seção, serão abordados os seguintes tópicos:

- Enquanto
- Faça-Enquanto
- Para

Laço Enquanto (Pré-Testado)

Se fosse necessário a elaboração de um jogo, como por exemplo um jogo da velha, e enquanto houvessem lugares disponíveis no tabuleiro, este jogo devesse continuar, como faríamos para que o algoritmo tivesse este comportamento? É simples. O comando **enquanto** poderia fazer esse teste lógico. A função do comando **enquanto** é: executar uma lista de comandos enquanto uma determinada condição for verdadeira.

A sintaxe é respectivamente a palavra reservada **enquanto**, a condição a ser testada entre parenteses, e entre chaves a lista de instruções que se deseja executar.

Exemplo de Sintaxe

```
01. logico condicao = verdadeiro
02. enquanto (condicao)
03. {
04.     //Executa a as instruções dentro do laço enquanto a condicao for verdadeira
05. }
```

A figura abaixo ilustra um algoritmo que verifica uma variável do tipo carácter. Enquanto a variável for diferente da letra 'S' o comando **enquanto** será executado, assim como as instruções dentro dele. No momento em que o usuário atribuir 'S' a variável, o comando enquanto terminará e o programa chega ao seu final.



O exemplo a seguir ilustra em portugol o mesmo algoritmo do fluxograma acima.

Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         caracter parar
06.         parar = 'N'
07.
08.         enquanto (parar != 'S')
09.         {
10.             escreva ("deseja parar o laço? (S/N)")
11.             leia (parar)
12.         }
13.     }
14. }
```

Laço Faça-Enquanto (Pós-Testado)

Em algumas situações, faz-se necessário verificar se uma condição é verdadeira ou não após uma entrada de dados do usuário. Para situações como essa, podemos usar o laço de repetição faca-enquanto. Este teste é bem parecido com o enquanto. A diferença está no fato de que o teste lógico é realizado no final, e com isso as instruções do laço sempre serão realizadas pelo menos uma vez. O teste verifica se elas devem ser repetidas ou não.

A sintaxe é respectivamente a palavra reservada **faca**, entre chaves as instruções a serem executadas, a palavra reservada **enquanto** e entre parenteses a condição a ser testada.

Exemplo de Sintaxe

```
01. logico condicao = verdadeiro
02. faca
03. {
04.     //Executa os comandos pelo menos uma vez, e continua executando enquanto a condição for verdadeira
05. } enquanto (condicao)
```

A figura abaixo ilustra um algoritmo que calcula a área de um quadrado. Note que para o cálculo da área é necessário que o valor digitado pelo usuário para aresta seja maior que 0. Caso o usuário informe um valor menor ou igual a 0 para a aresta, o programa repete o comando pedindo para que o usuário entre novamente com um valor para a aresta. Caso seja um valor válido, o programa continua sua execução normalmente e ao fim exibe a área do quadrado.



Exemplo

```
01. programa
02. {
03.     funcao inicio()
04.     {
05.         real aresta, area
06.
07.         faca
08.         {
```

```
09.             escreva ("Informe o valor da aresta: ")
10.             leia (aresta)
11.         } enquanto (aresta <= 0)
12.
13.         area=aresta*aresta
14.         escreva("A área é: ", area)
15.     }
16. }
```

Laço Para (Com Variável de Controle)

E se houver um problema em que sejam necessárias um número determinado de repetições? Por exemplo, se quiséssemos pedir ao usuário que digitasse 10 valores. Poderíamos utilizar a instrução Leia repetidas vezes. Porém se ao invés de 10 valores precisássemos de 100, essa tarefa se tornaria muito extensa. Para resolver problemas como esse, podemos usar um laço de repetição com variável de controle. No portugol, ele é conhecido como **para**.

O laço de repetição com variável de controle facilita a construção de algoritmos com número definido de repetições, pois possui um contador (variável de controle) embutido no comando com o incremento automático. Desta forma, um erro muito comum que se comete ao esquecer de fazer o incremento do contador é evitado. Toda vez que temos um problema cuja solução necessita de um número determinado de repetições utilizamos um contador. O contador deve ser inicializado antes do laço e deve ser incrementado dentro do laço.

O laço com variável de controle possui três partes. A inicialização da variável contadora, a definição do valor final do contador e a definição do incremento. Estas três partes são escritas juntas, no início do laço.

A sintaxe é respectivamente a palavra reservada **para**, abre parenteses, a declaração de uma variável de controle, ponto e virgula, a condição a ser testada, ponto e virgula, uma alteração na variável de controle a ser feita a cada iteração, fecha parenteses, e entre chaves as instruções do programa.

Exemplo de Sintaxe

```
01. para (inteiro i = 0; i < 8; i++)
02. {
03.     //Codigo a ser executado enquanto a condição for satisfeita.
04. }
```

A figura abaixo ilustra um algoritmo que exibe na tela a tabuada de 3. Note que conforme a sintaxe mostrada anteriormente, a primeira instrução do laço é inicializar o contador c=1. O segundo comando especifica a condição para que o laço continue a ser executado, ou seja, enquanto o contador c for menor ou igual a 10. Por último, a terceira instrução demonstra que o contador c será acrescentado em 1 em seu valor a cada iteração do comando. O laço será executado 10 vezes e mostrará a tabuada de 3.



O exemplo a seguir ilustra em portugol o mesmo algoritmo do fluxograma acima.

Exemplo

```
01. programa
02. {
```

```
03.     funcao inicio()
04.     {
05.         inteiro tab
06.
07.         para (inteiro c=1; c<=10; c++)
08.         {
09.             tab=c*3
10.             escreva ("3 x ", c, " = ", tab, "\n")
11.         }
12.     }
13. }
```

Bibliotecas

Em todo o algoritmo que se possa elaborar, existe a possibilidade da utilização de um conjunto de funções e comandos já existentes. A estes conjuntos de funções e comandos, dá-se o nome de Bibliotecas.

As bibliotecas contém códigos e dados auxiliares, que provém serviços a programas independentes, o que permite o compartilhamento e a alteração de código e dados de forma modular. Existem diversos tipos de bibliotecas, cada uma com funções para atender a determinados problemas.

Para se utilizar uma biblioteca é necessário primeiro importa-la para o seu programa.

No portugol para importar uma biblioteca usa-se as palavras reservadas **inclua biblioteca** seguido do nome da biblioteca que se deseja usar, e opcionalmente pode-se atribuir um apelido a ela usando do operador "-->" sem aspas seguido do apelido.

Para usar um recurso da biblioteca deve-se escrever o nome da biblioteca (ou apelido), seguido por um ponto e o nome do recurso a ser chamado como demonstrado abaixo.

Exemplo de Sintaxe

```
01. inclua biblioteca Mouse
02. inclua biblioteca Graficos --> g
03. Mouse.ocultar_cursor()
04. g.iniciar_modo_grafico(verdadeiro)
```

No portugol, existem as seguintes bibliotecas:

- Arquivos
- Graficos
- Matematica
- Mouse
- Sons
- Teclado
- Texto
- Tipos
- Util

Exemplo

```
01. programa
02. {
03.     inclua biblioteca Matematica
04.     inclua biblioteca Texto --> t
05.     funcao inicio()
06.     {
07.         real resultado
08.         resultado = Matematica.arredondar(Matematica.PI,5)
09.         escreva(resultado)
10.         escreva(t.caracteres_maiusculos("texto"))
11.     }
12. }
```